

Pretende-se uma aplicação para apoiar o departamento de contabilidade de uma empresa. O número de funcionários varia ao longo do tempo mas não é superior a 50. Para cada funcionário é registado o seu nome e vencimento. O programa deve ser orientado por um menu que ofereça as seguintes funcionalidades:

- a) Inserir funcionário;
- b) Listar todos os funcionários;
- c) Atualizar vencimento de um funcionário;
- d) Eliminar um funcionário;
- e) Mostrar informação ordenada por Nome de funcionário.

ED:

```
CONST TAMANHO=50
nomes[TAMANHO] TEXTO
vencimentos[TAMANHO], nElems, op INTEIRO
```

INICIO

```
nElems←0
```

REPETIR

```
op←menu()
```

CASO (op) SEJA

0:

```
    ESCREVER("Fim...")
```

1:

```
    SE (nElems<TAMANHO) ENTAO
```

```
        inserir(nomes, vencimentos, nElems)
```

```
        nElems←nElems+1
```

```
    SENAO
```

```
        ESCREVER("Atingida a capacidade máxima do vetor")
```

```
    FIMSE
```

2:

```
    SE (nElems>0) ENTAO
```

```
        listar(nomes, vencimentos, nElems)
```

```
    SENAO
```

```
        ESCREVER("Não existem funcionários")
```

```
    FIMSE
```

3:

```
    SE (nElems>0) ENTAO
```

```
        atualizarVencimento(nomes, vencimentos, nElems)
```

```
    SENAO
```

```
        ESCREVER("Não existem funcionários")
```

```
    FIMSE
```

4:

```
    SE (nElems>0) ENTAO
```

```
        SE (eliminar(nomes, vencimentos, nElems)=VERDADEIRO) ENTAO
```

```
            ESCREVER("Funcionário eliminado ")
```

```
            nElems←nElems-1
```

```
        SENAO
```

```
            ESCREVER("Não há qualquer funcionário com o nome introduzido ")
```

```
        FIMSE
```

```
    SENAO
```

```
        ESCREVER("Não existem funcionários")
```

```
    FIMSE
```

5:

```
    SE (nElems>0) ENTAO
```

```
        ordenar(nomes, vencimentos, nElems)
```

```
        listar(nomes, vencimentos, nElems)
```

```
    SENAO
```

```
        ESCREVER("Não existem funcionários")
```

```
    FIMSE
```

OUTRO:
ESCREVER(“Opção inválida”)

FIMCASO
ENQUANTO (op≠0)

FIM

```
DEFINIR INTEIRO menu()
ED:
    op INTEIRO
INICIO
    ESCREVER("0 – Sair")
    ESCREVER("1 – Inserir Funcionário")
    ESCREVER("2 – Listar funcionários")
    ESCREVER("3 – Atualizar vencimento de um funcionário")
    ESCREVER("4 – Eliminar funcionário")
    ESCREVER("5 – Mostrar informação ordenada por nome de funcionário")
    LER(op)
    RETORNA op
FIMDEFINIR
```

```
DEFINIR inserir(nomes[] TEXTO, vencimentos[] INTEIRO, nElems INTEIRO)
ED:
INICIO
    ESCREVER("Nome?")
    LER(nomes[nElems])
    ESCREVER("Vencimento?")
    LER(vencimentos[nElems])
FIMDEFINIR
```

```
DEFINIR listar(nomes[] TEXTO, vencimentos[] INTEIRO, nElems INTEIRO)
ED:
    x INTEIRO
INICIO
    ESCREVER("Listagem de funcionários")
    PARA (x ← 0 ATÉ (nElems-1) PASSO 1) FAZER
        ESCREVER(nomes[x], " - ", vencimentos[x])
    FIMPARA
FIMDEFINIR
```

```
DEFINIR INTEIRO pesquisar(nomes[] TEXTO, nElems INTEIRO, nome TEXTO)
ED:
    pos INTEIRO
INICIO
    pos ← 0
    ENQUANTO(pos < nElems E nome ≠ nomes[pos])
        pos ← pos + 1
    FIMENQUANTO
    SE (pos < nElems) ENTÃO
        RETORNA pos
    SENÃO
        RETORNA -1
    FIMSE
FIMDEFINIR
```

```

DEFINIR atualizarVencimento(nomes[] TEXTO, vencimentos[] INTEIRO, nElems INTEIRO)
ED:
    nome TEXTO
    pos INTEIRO
INICIO
    ESCREVER("Qual o nome do funcionário cujo vencimento pretende alterar")
    LER(nome)
    pos ← pesquisar(nomes, nElems, nome)
    SE (pos ≠ -1) ENTAO
        ESCREVER("Qual o novo vencimento")
        LER(vencimentos[pos])
    SENAO
        ESCREVER("Não há qualquer funcionário com o nome introduzido ")
    FIMSE
FIMDEFINIR

```

```

DEFINIR booleano eliminar(nomes[] TEXTO, vencimentos[] INTEIRO, nElems INTEIRO)
ED:
    nome TEXTO
    pos, x INTEIRO
INICIO
    ESCREVER("Qual o nome do funcionário que pretende eliminar")
    LER(nome)
    pos ← pesquisar(nomes, nElems, nome)
    SE (pos ≠ -1) ENTAO
        PARA (x ← pos ATÉ (nElems-2) PASSO 1) FAZER
            nomes[x] ← nomes[x+1]
            vencimentos[x] ← vencimentos[x+1]
        FIMPARA
        RETORNA VERDADEIRO
    SENAO
        RETORNA FALSO
    FIMSE
FIMDEFINIR

```

```

DEFINIR ordenar(nomes[] TEXTO, vencimentos[] INTEIRO, nElems INTEIRO)
ED:
    x, y, aux1 INTEIRO
    aux2 TEXTO
INICIO
    PARA (x ← 0 ATÉ (nElems-2) PASSO 1) FAZER
        PARA (y ← x+1 ATÉ (nElems-1) PASSO 1) FAZER
            SE (nomes[y] < nomes[x]) ENTAO
                aux2 ← nomes[x]
                nomes[x] ← nomes[y]
                nomes[y] ← aux2
                aux1 ← vencimentos[x]
                vencimentos [x] ← vencimentos [y]
                vencimentos [y] ← aux1
            FIMSE
        FIMPARA
    FIMPARA
FIMDEFINIR

```